# Example of a function constrained using Lagrange Multiplier

```
> restart:
> f  := (x,y,z) -> x+y+z ;
```

$$f := (x, y, z) \rightarrow x + y + z \quad (1)$$

```
> h1 := (x,y,z) -> z-2*x^2-y^2 ;
  h2 := (x,y,z) -> z-2*x-y^2 ;
```

$$h1 := (x, y, z) \rightarrow z - 2\,x^2 - y^2$$

$$h2 := (x, y, z) \rightarrow z - 2\,x - y^2 \quad (2)$$

```
> plot3d( [2*x^2+y^2,2*x+y^2*2], x=-20..20, y=-20..20 ) ;
```



```
> solve( [h1(x,y,z),h2(x,y,z)], [x,y,z] ) ;
```
$$[[x=0, y=y, z=y^2], [x=1, y=y, z=2+y^2]] \quad (3)$$

# Solve the minimization problem using parametrization (elimination) of the constraints

## First curve which eliminate constraints

Using the first parametrization

```
> PARA := x=0, z=y^2 ;
```
$$PARA := x=0, z=y^2 \quad (1.1.1)$$

```
> subs(PARA,h1(x,y,z) ) ;
  subs(PARA,h2(x,y,z) ) ;
```
$$0$$
$$0 \quad (1.1.2)$$

The function along the first constraint curve

```
> freduced := subs(PARA,f(x,y,z) ) ;
```
$$freduced := y+y^2 \quad (1.1.3)$$

Point(s) with null gradient

```
> diff(freduced,y) ; SOLy := solve( %, {y} );
```
$$1+2y$$
$$SOLy := \left\{y = -\frac{1}{2}\right\} \quad (1.1.4)$$

Hessian along the constraint

```
> subs( SOLy, diff(freduced,y,y) ) ;
```
$$2 \quad (1.1.5)$$

```
> subs(SOLy,PARA) ;
```
$$z = \frac{1}{4} \quad (1.1.6)$$

The point x=0, y=-1/2 and z =1/4 is a minimum point

## Second curve which eliminate constraints

Using the first parametrization

```
> PARA := x=1, z=2+y^2 ;
```
$$PARA := x=1, z=2+y^2 \quad (1.2.1)$$

```
> subs(PARA,h1(x,y,z) ) ;
  subs(PARA,h2(x,y,z) ) ;
```
$$0$$
$$0 \quad (1.2.2)$$

The function along the first constraint curve

```
> freduced := subs(PARA,f(x,y,z) ) ;
```

$$freduced := 3 + y + y^2 \tag{1.2.3}$$

Point(s) with null gradient

```
> diff(freduced,y) ; SOLy := solve( %, {y} );
```

$$1 + 2y$$

$$SOLy := \left\{y = -\frac{1}{2}\right\} \tag{1.2.4}$$

Hessian along the constraint

```
> subs( SOLy, diff(freduced,y,y) ) ;
```

$$2 \tag{1.2.5}$$

```
> subs(SOLy,PARA) ;
```

$$z = \frac{9}{4} \tag{1.2.6}$$

The point x=1, y=-1/2 and z =1/4 is a minimum point

# Solve the same problem using Lagrange Multiplier

Build the non linear system of the Lagrange multiplier

```
> gradf := [ diff(f(x,y,z),x), diff(f(x,y,z),y), diff(f(x,y,z),z) ]
  ;
```

$$gradf := [1, 1, 1] \tag{2.1}$$

```
> gradh1 := [ diff(h1(x,y,z),x), diff(h1(x,y,z),y), diff(h1(x,y,z),
  z) ] ;
  gradh2 := [ diff(h2(x,y,z),x), diff(h2(x,y,z),y), diff(h2(x,y,z),
  z) ] ;
```

$$gradh1 := [-4x, -2y, 1]$$

$$gradh2 := [-2, -2y, 1] \tag{2.2}$$

The non linear system for the max/min

```
> EQ1 := gradf[1] - lambda1 * gradh1[1] - lambda2 * gradh2[1] = 0 ;
  EQ2 := gradf[2] - lambda1 * gradh1[2] - lambda2 * gradh2[2] = 0 ;
  EQ3 := gradf[3] - lambda1 * gradh1[3] - lambda2 * gradh2[3] = 0 ;
  EQ4 := h1(x,y,z) = 0 ;
  EQ5 := h2(x,y,z) = 0 ;
```

$$EQ1 := 1 + 4\lambda1\, x + 2\lambda2 = 0$$

$$EQ2 := 1 + 2\lambda1\, y + 2\lambda2\, y = 0$$

$$EQ3 := 1 - \lambda1 - \lambda2 = 0$$

$$EQ4 := z - 2x^2 - y^2 = 0$$

$$EQ5 := z - 2x - y^2 = 0 \tag{2.3}$$

```
> solve( {EQ||(1..5)}, {x,y,z,lambda1,lambda2} ) ;
```

$$\left\{\lambda1 = \frac{3}{2}, \lambda2 = -\frac{1}{2}, x = 0, y = -\frac{1}{2}, z = \frac{1}{4}\right\}, \left\{\lambda1 = -\frac{3}{2}, \lambda2 = \frac{5}{2}, x = 1, y = -\frac{1}{2}, z = \frac{9}{4}\right\} \tag{2.4}$$

```
> SOLx := solve( EQ1, {x} ) ;
  SOLy := solve( EQ2, {y} ) ;
```

$$SOLx := \left\{x = -\frac{1}{4}\,\frac{1+2\,\lambda 2}{\lambda 1}\right\}$$

$$SOLy := \left\{y = -\frac{1}{2\,(\lambda 1+\lambda 2)}\right\} \tag{2.5}$$

```
> SOLy := algsubs( EQ3, SOLy ) ;
```

$$SOLy := \left\{y = -\frac{1}{2}\right\} \tag{2.6}$$

```
> EQ4bis := subs( SOLy, EQ4 ) ;
  EQ5bis := subs( SOLy, EQ5 ) ;
```

$$EQ4bis := z - 2\,x^2 - \frac{1}{4} = 0$$

$$EQ5bis := z - 2\,x - \frac{1}{4} = 0 \tag{2.7}$$

```
> solve( EQ5bis, {z} ); EQ4tris := subs( %, EQ4bis ) ;
```

$$\left\{z = 2\,x + \frac{1}{4}\right\}$$

$$EQ4tris := 2\,x - 2\,x^2 = 0 \tag{2.8}$$

```
> solve( EQ4tris, {x} ) ;
```

$$\{x = 0\},\ \{x = 1\} \tag{2.9}$$

## First solution

```
> SOL1 := op(SOLy), x=0 ;
```

$$SOL1 := y = -\frac{1}{2},\ x = 0 \tag{2.1.1}$$

```
> subs( SOL1, [EQ1,EQ2,EQ3,EQ4,EQ5] ) ;
```

$$\left[1+2\,\lambda 2 = 0,\ 1-\lambda 1-\lambda 2 = 0,\ 1-\lambda 1-\lambda 2 = 0,\ z-\frac{1}{4} = 0,\ z-\frac{1}{4} = 0\right] \tag{2.1.2}$$

```
> SOL1 := SOL1, z=1/4 ;
```

$$SOL1 := y = -\frac{1}{2},\ x = 0,\ z = \frac{1}{4} \tag{2.1.3}$$

```
> subs( SOL1, [EQ1,EQ2,EQ3,EQ4,EQ5] ) ; solve(%[1],{lambda2} );
```

$$[1+2\,\lambda 2 = 0,\ 1-\lambda 1-\lambda 2 = 0,\ 1-\lambda 1-\lambda 2 = 0,\ 0 = 0,\ 0 = 0]$$

$$\left\{\lambda 2 = -\frac{1}{2}\right\} \tag{2.1.4}$$

```
> SOL1 := SOL1, lambda2=-1/2 ;
```

$$SOL1 := y = -\frac{1}{2},\ x = 0,\ z = \frac{1}{4},\ \lambda 2 = -\frac{1}{2} \tag{2.1.5}$$

```
> subs( SOL1, [EQ1,EQ2,EQ3,EQ4,EQ5] ) ; solve(%[2],{lambda1} );
```

$$\left[0 = 0,\ \frac{3}{2}-\lambda 1 = 0,\ \frac{3}{2}-\lambda 1 = 0,\ 0 = 0,\ 0 = 0\right]$$

(2.1.6)

$$\left\{\lambda 1 = \frac{3}{2}\right\} \quad (2.1.6)$$

```
> SOL1 := SOL1, lambda1=3/2 ;
```

$$SOL1 := y = -\frac{1}{2}, x = 0, z = \frac{1}{4}, \lambda 2 = -\frac{1}{2}, \lambda 1 = \frac{3}{2} \quad (2.1.7)$$

```
> subs(SOL1,[EQ||(1..5)] ) ;
```

$$[0 = 0, 0 = 0, 0 = 0, 0 = 0, 0 = 0] \quad (2.1.8)$$

## Second solution

```
> SOL2 := op(SOLy), x=1 ;
```

$$SOL2 := y = -\frac{1}{2}, x = 1 \quad (2.2.1)$$

```
> subs( SOL2, [EQ1,EQ2,EQ3,EQ4,EQ5] ) ;
```

$$\left[1 + 4\lambda 1 + 2\lambda 2 = 0, 1 - \lambda 1 - \lambda 2 = 0, 1 - \lambda 1 - \lambda 2 = 0, z - \frac{9}{4} = 0, z - \frac{9}{4} = 0\right] \quad (2.2.2)$$

```
> SOL2 := SOL2, z=9/4 ;
```

$$SOL2 := y = -\frac{1}{2}, x = 1, z = \frac{9}{4} \quad (2.2.3)$$

```
> subs( SOL2, [EQ1,EQ2,EQ3,EQ4,EQ5] ) ; solve(%[1..2],{lambda1,
lambda2} );
```

$$[1 + 4\lambda 1 + 2\lambda 2 = 0, 1 - \lambda 1 - \lambda 2 = 0, 1 - \lambda 1 - \lambda 2 = 0, 0 = 0, 0 = 0]$$

$$\left\{\lambda 1 = -\frac{3}{2}, \lambda 2 = \frac{5}{2}\right\} \quad (2.2.4)$$

```
> SOL2 := SOL2, lambda1 = -3/2, lambda2 = 5/2 ;
```

$$SOL2 := y = -\frac{1}{2}, x = 1, z = \frac{9}{4}, \lambda 1 = -\frac{3}{2}, \lambda 2 = \frac{5}{2} \quad (2.2.5)$$

```
> subs(SOL2,[EQ||(1..5)] ) ;
```

$$[0 = 0, 0 = 0, 0 = 0, 0 = 0, 0 = 0] \quad (2.2.6)$$

```
> #solve({EQ1,EQ2,EQ3},{x,y,lambda} ) ; allvalues(%) ;
```

Find the space of vectors that are in the kernel of the contraint h(x,y)

```
> with(LinearAlgebra) :
```

Build the hessian

```
> hessf := <<diff(f(x,y,z),x,x),diff(f(x,y,z),x,y),diff(f(x,y,z),x,z)
>
         |<diff(f(x,y,z),y,x),diff(f(x,y,z),y,y),diff(f(x,y,z),y,z)
>
         |<diff(f(x,y,z),z,x),diff(f(x,y,z),z,y),diff(f(x,y,z),z,z)
>> ;
```

$$hessf := \begin{bmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{bmatrix} \quad (4)$$

```
> hessh1 := <<diff(h1(x,y,z),x,x),diff(h1(x,y,z),x,y),diff(h1(x,y,z),
x,z)>
```

```
              |<diff(h1(x,y,z),y,x),diff(h1(x,y,z),y,y),diff(h1(x,y,z),
  y,z)>
              |<diff(h1(x,y,z),z,x),diff(h1(x,y,z),z,y),diff(h1(x,y,z),
  z,z)>> ;
```

$$hessh1 := \begin{bmatrix} -4 & 0 & 0 \\ 0 & -2 & 0 \\ 0 & 0 & 0 \end{bmatrix} \quad (5)$$

```
> hessh2 := <<diff(h2(x,y,z),x,x),diff(h2(x,y,z),x,y),diff(h2(x,y,z),
  x,z)>
              |<diff(h2(x,y,z),y,x),diff(h2(x,y,z),y,y),diff(h2(x,y,z),
  y,z)>
              |<diff(h2(x,y,z),z,x),diff(h2(x,y,z),z,y),diff(h2(x,y,z),
  z,z)>> ;
```

$$hessh2 := \begin{bmatrix} 0 & 0 & 0 \\ 0 & -2 & 0 \\ 0 & 0 & 0 \end{bmatrix} \quad (6)$$

```
> hessL1 := hessf - lambda1 * hessh1 - lambda2 * hessh2;
```

$$hessL1 := \begin{bmatrix} 4\,\lambda 1 & 0 & 0 \\ 0 & 2\,\lambda 1 + 2\,\lambda 2 & 0 \\ 0 & 0 & 0 \end{bmatrix} \quad (7)$$

Analyse the first solution

```
> G1 := subs(SOL1,gradh1);
  G2 := subs(SOL1,gradh2);
```

$$G1 := [0, 1, 1]$$
$$G2 := [-2, 1, 1] \quad (8)$$

```
> G := Transpose(<<op(G1)>|<op(G2)>>);
```

$$G := \begin{bmatrix} 0 & 1 & 1 \\ -2 & 1 & 1 \end{bmatrix} \quad (9)$$

```
> G.<z1,z2,z3> ;
```

$$\begin{bmatrix} z2 + z3 \\ -2\,z1 + z2 + z3 \end{bmatrix} \quad (10)$$

```
> Z1 := alpha*op(NullSpace(G)) ;
```

$$Z1 := \begin{bmatrix} 0 \\ -\alpha \\ \alpha \end{bmatrix} \quad (11)$$

```
> hessL1 := subs( SOL1, hessL ) ;
```

$$hessL1 := hessL \quad (12)$$

The vector in the kernel of grad h1(x,y,z) and h2(x,y,z) and

```
> Transpose(Z1).hessL1.Z1 ;
```

$$\begin{bmatrix} 0 & -\alpha & \alpha \end{bmatrix} . hessL . \begin{bmatrix} 0 \\ -\alpha \\ \alpha \end{bmatrix} \quad (13)$$

Analyse the second solution

```
> G1 := subs(SOL2,gradh1);
  G2 := subs(SOL2,gradh2);
```

$$G1 := [-4, 1, 1]$$
$$G2 := [-2, 1, 1] \quad (14)$$

```
> G := Transpose(<<op(G1)>|<op(G2)>>);
```

$$G := \begin{bmatrix} -4 & 1 & 1 \\ -2 & 1 & 1 \end{bmatrix} \quad (15)$$

```
> G.<z1,z2,z3> ;
```

$$\begin{bmatrix} -4\, z1 + z2 + z3 \\ -2\, z1 + z2 + z3 \end{bmatrix} \quad (16)$$

```
> Z2 := alpha*op(NullSpace(G)) ;
```

$$Z2 := \begin{bmatrix} 0 \\ -\alpha \\ \alpha \end{bmatrix} \quad (17)$$

```
> hessL2 := subs( SOL2, hessL ) ;
```

$$hessL2 := hessL \quad (18)$$

The vector in the kernel of grad h1(x,y,z) and h2(x,y,z) and

```
> Transpose(Z2).hessL2.Z2 ;
```

$$\begin{bmatrix} 0 & -\alpha & \alpha \end{bmatrix} . hessL . \begin{bmatrix} 0 \\ -\alpha \\ \alpha \end{bmatrix} \quad (19)$$